

Extract from the Help pages:

ARROWS?

(in PostScript graphics only)

Points in a scatter plot can be connected by arrows (or line segments) by giving an ARROWS specification.

Its simplest form is

ARROWS=A

where A=[a_{ij}] is an incidence matrix saved as a matrix file and a_{ij}=1 means that points of observations i and j are to be connected and a_{ij}=0 that points i and j are not to be connected.

It is assumed that when the data set has n observations, A is an n x n matrix. Thus restrictions by IND, CASES etc. are not permitted.

The general form of ARROWS is

ARROWS=matrix,gap,type,alen,angle

where

matrix:	incidence matrix (MAT file)	default
gap:	distance from end points (in plotting units)	-
type:	0=no arrow, 1=end arrow, 2=start arrow, 3=both, 4= <-->	0
	corresponding negative value = filled arrow head	
alen:	length of arrow (in plotting units)	20
angle:	angle (sharpness) of arrow (in radians)	pi/6

To give a possibility to configure each connection individually (i.e. variation in line width, type, and color as well as in arrow type), the non-zero elements can be presented as decimal numbers of form

a_{ij} = swww.tacc

where www (one or more digits)	is the line width as multiple of	0.24 Points, 0.24(Point:mm)=0.084666...
t (one digit)	is index of line type (LINETYPE?),	
a (one digit)	is arrow type (see the list above),	
cc (two digits)	is the index of color given by FILL spec.,	
s (sign)	'-' gives filled arrow head(s).	

For example, a_{ij}=-5.0304 implies an arrow having width 5*0.24=1.2 Points, solid line type (0), arrow type >---> (3), filled arrow heads (-), and color (4) specified by FILL(-4)=c,m,y,k (See PSCOLOR?).

Examples:

.....

DATA Datal

Name X Y

P1 1 1

P2 1 2

P3 1.8 1.2

P4 2 2

Lines (arrows) defined by an incidence matrix:

MATRIX A

/// P1 P2 P3 P4

P1 0 1 0 1

P2 0 0 0 1


P3 0 1 0 0

P4 0 0 1 0


MAT SAVE A

Simple lines:


PLOT Datal,X,Y / ARROWS=A DEVICE=PS,A1.PS




.....
 Simple lines (gaps of 40 plotting units from each point)
 PLOT Datal,X,Y / ARROWS=A,40 DEVICE=PS,A2.PS POINT=Name




.....
 Arrow head at the end point:
 PLOT Datal,X,Y / ARROWS=A,40,1 DEVICE=PS,A3.PS POINT=Name




.....
 Arrow head at both ends (3), arrow side length 30 and angle pi/12:
 pi=3.141592653589793
 PLOT Datal,X,Y / ARROWS=A,40,3,30,pi/12 DEVICE=PS,A4.PS



.....
 "Line segments":
 pi=3.141592653589793
 PLOT Datal,X,Y / ARROWS=A,40,3,10,pi/2 DEVICE=PS,A6.PS POINT=Name




.....
 Filled arrow heads:
 PLOT Datal,X,Y / ARROWS=A,40,-1,30 DEVICE=PS,A7.PS POINT=Name



```

.....
Filled arrow heads at both ends:
PLOT Data1,X,Y / ARROWS=A,40,-3,30 DEVICE=PS,A8.PS POINT=Name


```



```

.....
Setting joint line width, type, and color:
DEVICE=PS,A9.PS POINT=Name
ARROWS=[line_width(1)][line_type(3)][RED],A,40,-3,30
PLOT Data1,X,Y


```



```

.....
arrow type 4: <----->
PLOT Data1,X,Y / ARROWS=A,40,4 DEVICE=PS,A10.PS

```



```

.....
Generalized incidence matrix:
MATRIX B
/// P1 P2 P3 P4
P1 0 1 0 x // x=-5.2101
P2 0 0 0 1
P3 x y 0 0 // y=-5.0401
P4 0 0 z 0 // z=1.0102


```

MAT SAVE B

```

FILL(-1)=0,1,1,0 FILL(-2)=0.5,0,0.5,0
PLOT Data1,X,Y / ARROWS=B,40,+3,30,50 DEVICE=PS,A11.PS

```



Correlation matrix as a vector diagram:

```

* .....
*The graph is automatically generated and displayed by activating:
*/ACTIVATE +
*
*Selecting variables in a proper order:
*MASK=--BCDEFHIJKLAG
*
+CORR DECA
+MAT LOAD CORR.M,11.1,CUR+1
*MATRIX CORR.M
*R(DECA)
*///      Heig 100m L_ju Shot Hi_j 400m Weig Hurd Disc Pole Jave 1500
*Height   1.0 -0.1 -0.0  0.6  0.1 -0.2  0.9  0.2  0.6 -0.4 -0.0 -0.2
*100m     -0.1  1.0  0.2 -0.0 -0.4  0.5 -0.1  0.3  0.0  0.1 -0.2 -0.3
*L_jump   -0.0  0.2  1.0 -0.0 -0.0  0.1 -0.1  0.3  0.0  0.1  0.2 -0.2
*Shot_put  0.6 -0.0 -0.0  1.0  0.2 -0.3  0.7  0.1  0.7 -0.2  0.0 -0.4
*Hi_jump   0.1 -0.4 -0.0  0.2  1.0 -0.3  0.2 -0.0  0.2 -0.1  0.1 -0.1
*400m     -0.2  0.5  0.1 -0.3 -0.3  1.0 -0.3  0.2 -0.3  0.0 -0.1  0.3
*Weight    0.9 -0.1 -0.1  0.7  0.2 -0.3  1.0  0.1  0.6 -0.3 -0.1 -0.4
*Hurdles   0.2  0.3  0.3  0.1 -0.0  0.2  0.1  1.0  0.0 -0.1 -0.1 -0.2
*Discus    0.6  0.0  0.0  0.7  0.2 -0.3  0.6  0.0  1.0 -0.2  0.1 -0.6
*Pole_vlt -0.4  0.1  0.1 -0.2 -0.1  0.0 -0.3 -0.1 -0.2  1.0 -0.1  0.0
*Javelin  -0.0 -0.2  0.2  0.0  0.1 -0.1 -0.1 -0.1  0.1 -0.1  1.0 -0.1
*1500m    -0.2 -0.3 -0.2 -0.4 -0.1  0.3 -0.4 -0.2 -0.6  0.0 -0.1  1.0
*
*Stepwise transformation of the correlation matrix
*to a suitable generalized incidence matrix:
*n=12
+MAT R=CORR.M
*Removing diagonal elements:
+MAT R=R-IDN(n,n)
*Multiplying correlations by 10 and rounding to the closest integer:
+MAT TRANSFORM R BY round(10*X#)
*Setting color 1 for positive correlations and 2 for negative:
+MAT #TRANSFORM R BY if(X#>0)then(abs(X#)+0.0001)else(abs(X#)+0.0002)
*Setting heads for arrows starting from 'external' variables Height and Weight:
+MAT #TRANSFORM R BY if(I#=1)then(X#+0.01)else(X#)
+MAT #TRANSFORM R BY if(I#=7)then(X#+0.01)else(X#)
*Setting 'too small' values to zero:
+MAT #TRANSFORM R BY if(X#<0.1)then(0)else(X#)
*Removing heads from arrows between external variables:
+MAT R(1,7)=R(1,7)-0.01
+MAT R(7,1)=0
*Converting to negative values (getting filled arrow heads):
+MAT R=(-1)*R
*
*(Data after activating two VAR operations below)
*DATA DECA12,A,A+11,N,M
*
MAAAAAAAAA 11.111 11.111      11.111 11.111
NEvent     X      Y          X'      Y'
*
AHeight    1.000  0.000      1.070  0.000
*100m      0.866  0.500      0.913  0.550
*L_jump    0.500  0.866      0.485  0.953
*Shot_put  0.000  1.000     -0.100  1.100
*Hi_jump   -0.500  0.866     -0.685  0.953
*400m     -0.866  0.500     -1.113  0.550
*Weight    -1.000  0.000     -1.270  0.000
*Hurdles   -0.866 -0.500     -1.113 -0.550
*Discus    -0.500 -0.866     -0.685 -0.953
*Pole_vlt -0.000 -1.000     -0.100 -1.100
*Javelin   0.500 -0.866      0.485 -0.953
*1500m     0.866 -0.500      0.913 -0.550
*
*Points on a regular polygon of n sides:
*pi=3.141592653589793
+VAR X,Y TO DECA12
*X=cos(2*(ORDER-1)/n*pi)
*Y=sin(2*(ORDER-1)/n*pi)
*
+PLOT DECA12,X,Y / POINT=3,20 ARROWS=R,50
*FILL(-1)=0,1,1,0 FILL(-2)=1,1,0,0


```

```

*DEVICE=PS,K1.PS SCALE=-1.5,1.5
*SIZE=1500,1500 XDIV=0,1,0 YDIV=0,1,0 HEADER= XLABEL= YLABEL= FRAME=0
* .....
*pi=3.141592653589793 n=12
*Larger ellipse for names with center moved to the left:
*R1=1.17 R2=1.1 C=0.1
+VAR X',Y' TO DECA12
*X'=R1*cos(2*(ORDER-1)/n*pi)-C
*Y'=R2*sin(2*(ORDER-1)/n*pi)
*
+PLOT DECA12,X',Y' / POINT=[Swiss(9)],Event
*DEVICE=PS,K2.PS SCALE=-1.5,1.5
*SIZE=1500,1500 XDIV=0,1,0 YDIV=0,1,0 XLABEL= YLABEL= FRAME=0
*HEADER=[Swiss(15)],Connections_in_Decathlon
* .....
+EPS JOIN Connections,K1,K2
+/GS-PDF Connections.PS
*

```

Connections in Decathlon



It is easy to modify this setup for other correlation and distance matrices.